

# FRAKZIOEN SINPLIFIKAZIOA



URREKO alearen ildoari jarraituz, jorra dezagun ale honetan frakzioen sinplifikazioa. Denok dakigunez, zatikia zenbaki razionala idazteko era bat da, eta  $a/b$  moduan adierazten da. Beraz, zatiki modura idatz daitezkeen zenbakiak, zenbaki razionalak dira.

Bestalde badakigu sinplifikazioa zatikien kalkulurako erregela bat dela. Erregela honek dioenez, eta besteak beste,  $3/5$  eta  $9/15$  frakzioek zatiki bera adierazten dute. Beraz, frakzio baten zenbakitzailea eta izendatzailea zero ez den zenbaki berdinean batez zatitzen badira, 3 da kasu

honetan zenbaki hori, frakzioa ez da aldatzen. Hots, frakzio berriek zatiki bera adierazten dute.

Honez gain, 33. alean azaldu genuenez, zenbaki arrunt batzuek aldi berean zatitzailetzat duten zenbakirik handiena da Zatitzaile Komunetako Handiena. Zatiki bateko zenbakitzailea eta izendatzailea, bestalde, zenbaki arruntak direnez, honako hau esan dezakegu: Zatitzaile Komunetako Handiena da aipatu zenbaki zatitzaile hori.

Frakzioen sinplifikazioari sarreratxo hau egin ondoren, programa azaltzeari ekingo diogu. Hona hemen horretarako alde zuzenetik frakzioen sinplifikazioa burutzen duen programa. Ea lagungarri zaizun.

```

10  REM ZATIPIAK SINPLIFIKATU
20  DIM Z1(10,2),Z2(10,2),DES(10,2),ZKH(10,2)
30  SCREEN 2:SCREEN 0:KEY OFF
40  RANDOMIZE TIMER
50  GEH$="B"
60  WHILE GEH$="B" OR GEH$="b"
70      FOR I=1 TO 10:Z1(I,1)=0:Z1(I,2)=0:Z2(I,1)=0:Z2(I,2)=0:NEXT I
80      ALE1=INT(RND*999)+1:ZEN=ALE1:GOSUB 1000
90      ZK1=J:FOR I=1 TO 10:Z1(I,1)=DES(I,1):Z1(I,2)=DES(I,2):NEXT I
100     ALE2=INT(RND*998)+2:ZEN=ALE2:GOSUB 1000
110     ZK2=J:FOR I=1 TO 10:Z2(I,1)=DES(I,1):Z2(I,2)=DES(I,2):NEXT I
120     REM  ZKHren kalkulua
130     FOR I=1 TO 10 :ZKH(I,1)=0:ZKH(I,2)=0:NEXT I
140     I=0:J1=1:J2=1
150     WHILE J1<=ZK1
160     WHILE (Z1(J1,1)>Z2(J2,1)) AND Z2(J2,1)<>0:J2=J2+1:WEND
170     IF Z1(J1,1)<>Z2(J2,1) THEN 200
180     I=I+1:ZKH(I,1)=Z1(J1,1)
190     IF Z1(J1,2)<=Z2(J2,2) THEN ZKH(I,2)=Z1(J1,2)
                                     ELSE ZKH(I,2)=Z2(J2,2)
200     J1=J1+1:J2=1
210     WEND
220     ZKH=1:ZK3=I
230     FOR JJ=1 TO I:ZKH=(ZKH(JJ,1)^ZKH(JJ,2))*ZKH:NEXT JJ
240     REM AMAIA ZKHren kalkulua
250     ZENB=ALE1/ZKH:IZEND=ALE2/ZKH
260     CLS:COLOR 10:LOCATE 1,31:PRINT "ZATIPIEN SINPLIFIKAZIOA":COLOR 11:
        LOCATE 2,31:PRINT STRING$(23,61)
    
```

```

270 LOCATE 5,5:COLOR 2:PRINT "Simplifika ezazu ondoko zatikia:"
280 LOCATE 8,16:COLOR 13:PRINT "_____";:COLOR 10:PRINT "= ":COLOR 13:
LOCATE 7,16:PRINT ALE1:LOCATE 9,16:PRINT ALE2
290 COLOR 10:LOCATE 8,24:PRINT "_____ "
300 LOCATE 23,5:COLOR 3:PRINT "Sar ezazu zenbakitzailea eta <RETURN>
sakatu.":COLOR 10
310 LOCATE 7,24:INPUT " ",ZENB1$:ZENB1=VAL(ZENB1$):LOCATE 23,1:
PRINT SPC(79)
320 LOCATE 23,5:COLOR 3:PRINT "Sar ezazu izendatzailea eta <RETURN> sakatu":
COLOR 10
330 LOCATE 9,24:INPUT " ",IZEND1$:IZEND1=VAL(IZEND1$):LOCATE 23,1:
PRINT SPC(79)
340 LOCATE 23,20:COLOR 11
350 IF ZENB=ZENB1 AND IZEND=IZEND1 THEN 370
360 PRINT "GAIZKI!!! Simplifikazioa gaizki egin duzu.":SOUND 300,4:
SOUND 150,10:GOTO 380
370 PRINT "ONGI!!! Simplifikazioa ongi egin duzu.":
FOR I=1 TO 10:SOUND 200*I,.5:NEXT I
380 FOR JJ=1 TO 3000:NEXT JJ
390 GOSUB 2000
400 COLOR 13:LOCATE 14,5:COLOR 13:PRINT "_____";:COLOR 11:PRINT "= ":
COLOR 13:LOCATE 13,5:PRINT ALE1:LOCATE 15,5:PRINT ALE2
410 COLOR 11:LOCATE 14,13:PRINT STRING$(LUZ+2,"_");" = ":X=POS(0)
420 XX=13:GOSUB 3000
430 FOR II=1 TO ZK3
440 J1=1
450 WHILE Z1(J1,1)<>ZKH(II,1):J1=J1+1:WEND
460 Z1(J1,2)=Z1(J1,2)-ZKH(II,2)
470 IF Z1(J1,2)>0 THEN 520
480 FOR JJ=J1 TO ZK1-1
490 Z1(JJ,1)=Z1(JJ+1,1):Z1(JJ,2)=Z1(JJ+1,2)
500 NEXT JJ
510 Z1(ZK1,1)=0:ZK1=ZK1-1
520 J2=1
530 WHILE Z2(J2,1)<>ZKH(II,1):J2=J2+1:WEND
540 Z2(J2,2)=Z2(J2,2)-ZKH(II,2)
550 IF Z2(J2,2)>0 THEN 600
560 FOR JJ=J2 TO ZK2-1
570 Z2(JJ,1)=Z2(JJ+1,1):Z2(JJ,2)=Z2(JJ+1,2)
580 NEXT JJ
590 Z2(ZK2,1)=0:ZK2=ZK2-1
600 NEXT II
610 IF ZKH=1 THEN 680
620 IF ZK1=0 AND ZK2=0 THEN COLOR 13:LOCATE 14,18+LUZ:PRINT "1":GOTO 700
630 XX=18+LUZ:GOSUB 2000
640 IF ZK2=0 THEN 660
650 LOCATE 14,XX:PRINT STRING$(LUZ+2,"_");" = ":X=POS(0)
660 GOSUB 3000
670 IF IZEND=1 THEN PRINT "= ":X=POS(0):COLOR 13
680 IF IZEND=1 THEN LOCATE 14,X+2:PRINT ZENB:GOTO 700
690 COLOR 13:LOCATE 14,X:PRINT "_____":LOCATE 13,X:PRINT ZENB:
LOCATE 15,X:PRINT IZEND
700 LOCATE 23,1:PRINT SPC(79);
710 COLOR 11:LOCATE 23,19:INPUT "Beste ariketarik egin nahi al duzu (B/E)?",GEH$
720 WEND
730 CLS
740 END

```

```

750 REM REM
1000 REM DESKONPOSATU
1010 FOR I=1 TO 10:DES(I,1)=0:DES(I,2)=0:NEXT I
1020 J=0:I=2
1030 WHILE I<=INT(ZEN/2)
1040     KONT=0
1050     WHILE ZEN-INT(ZEN/I)*I=0
1060         KONT=KONT+1:ZEN=INT(ZEN/I)
1070     WEND
1080     IF KONT>0 THEN J=J+1:DES(J,1)=I:DES(J,2)=KONT
1090     I=I+1
1100 WEND
1110 IF J=0 THEN J=1:DES(J,1)=ZEN:DES(J,2)=1:GOTO 1130
1120 IF ZEN<>DES(J,1) AND ZEN>1 THEN J=J+1:DES(J,1)=ZEN:DES(J,2)=1
1130 RETURN
1140 REM AMAIA DESKONPOSATU
1150 REM REM
2000 REM LUZERA
2010 LUZ1=0
2020 FOR II=1 TO ZK1
2030     LUZ1=LUZ1+LEN(STR$(Z1(II,1)))-1
2040     IF Z1(II,2)>1 THEN LUZ1=LUZ1+LEN(STR$(Z1(II,2)))+1
2050     IF Z1(II+1,1)>0 THEN LUZ1=LUZ1+3
2060 NEXT II
2070 LUZ2=0
2080 FOR II=1 TO ZK2
2090     LUZ2=LUZ2+LEN(STR$(Z2(II,1)))-1
2100     IF Z2(II,2)>1 THEN LUZ2=LUZ2+LEN(STR$(Z2(II,2)))+1
2110     IF Z2(II+1,1)>0 THEN LUZ2=LUZ2+3
2120 NEXT II
2130 IF LUZ1>LUZ2 THEN LUZ=LUZ1 ELSE LUZ=LUZ2
2140 RETURN
2150 REM AMAIA LUZERA
2160 REM REM
3000 REM IDATZI
3010 IF ZK2=0 THEN L=14 ELSE L=13
3020 LOCATE 13,XX+INT((LUZ-LUZ1)/2)
3030 FOR II=1 TO ZK1
3040     LOCATE L,POS(0):PRINT Z1(II,1);
3050     IF Z1(II,2)>1 THEN LOCATE L-1,POS(0)-1:PRINT Z1(II,2);
3060     IF Z1(II+1,1)>0 THEN LOCATE L,POS(0):PRINT “.”;
3070 NEXT II
3080 IF ZK1=0 THEN LOCATE 13,XX+INT((LUZ-1)/2):PRINT “1”;
3090 ESP$=”E”
3100 FOR II=1 TO ZK2
3110     IF Z2(II,2)>1 THEN ESP$=”B”
3120 NEXT II
3130 IF ESP$=”B” THEN YY=16 ELSE YY=15
3140 IF ZK2>0 THEN LOCATE YY,XX+INT((LUZ-LUZ2)/2)
3150 FOR II=1 TO ZK2
3160     LOCATE YY,POS(0):PRINT Z2(II,1);
3170     IF Z2(II,2)>1 THEN LOCATE YY-1,POS(0)-1:PRINT Z2(II,2);
3180     IF Z2(II+1,1)>0 THEN LOCATE YY,POS(0):PRINT “.”;
3190 NEXT II
3200 RETURN
3210 REM AMAIA IDATZI

```

## PROGRAMAREN AZALPENA

- 20-50: Programaren hasieraketa.  
60-720: Programaren ziklo nagusia.
- 70-110: Bi zenbaki aleatorio atera eta faktoretan deskonposatu (GOSUB 1000).
- 120-240: Zatitzaile Komunetako Handienaren (ZKHren) kalkulua. Zehaztasun gehiagorako, ikus *Elhuyar. Zientzia eta Teknika* aldizkariaren 33. alea.
- 250: Frakzio sinplifikatuaren zenbakitzaile (ZENB) eta izendatzailearen (IZEND) kalkulua.
- 260-290: Sortutako frakzioaren idazketa pantailan.
- 300-340: Erabiltzaileak sartutako erantzuna jaso.
- 350-380: Sartutako erantzuna zuzena den ala ez egiaztatu eta erabiltzaileari adierazi.
- 390-700: Emaizta zuzena agertzen da sinplifikazioa urratsez urrats burutuz.

Adibidez:

$$\frac{36}{756} = \frac{2^3 \cdot 3^2 \cdot 5}{2^2 \cdot 3^3 \cdot 7} = \frac{2 \cdot 5}{3 \cdot 7} = \frac{10}{21}$$

$$Z.K.H. = 2^2 \cdot 3^2$$

- 390-420: Hasierako frakzioa idazten da.
- 430-600: Emandako frakzio deskonposatua, zenbakitzailea (440-510) batetik eta izendatzailea (520-590) bestetik, ZKHk duen faktore bakoitzaz zatitzen da. Horretarako ZKHren faktore bakoitza zenbakitzailean eta izendatzailean aurkitzen da. Aurkitu ondoren, berretzaileen kenketa burutzen da emaitzaren faktorea ekoiztuz. Berretzaileen arteko kendura 0 den kasuan, faktorea eliminatu egiten da (480-500 eta 560-580).
- 610-700: Burutu diren kalkuluen idazketa.

## PROZEDURAK

- 1000-1140: Zenbaki bat faktoretan deskonposatzeko prozedura. Ikus *Elhuyar. Zientzia eta Teknika* aldizkariaren 31. alea.
- 2000-2050: Faktoretan deskonposaturik dagoen zatikiaren lerroaren luzera kalkulatzeko prozedura. Zenbakitzailearen eta izendatzailearen luzera kalkulatu da eta bietan luzeena hartzen.
- 3000-3210: Faktoretan deskonposaturik dagoen zatikia idazteko prozedura. ■

**EUSKAL  
ADITZ  
LAGUNTZAILEA**

**(Softwarea)**

*Ordenadorea erabiliz,  
aditzarekin jolastuz, milaka  
ariketa egin ditzakezu*

**JADANIK  
KALEAN**